

GB



ology@github

Search this website .

Imitating Steve Reich

2020-10-18 BY GENE



Recently I was enjoying Steve Reich's "[Octet](#)" ([Eight Lines](#)) and became curious if I could mimic the simple, staccato phrases with which he starts off his piece. tl;dr: [reichify](#).

Fortunately I have been immersed in algorithmic composition of late and had a fair idea what tools I would need. Number one is [Perl](#) and its vast ecosystem of modules – music ones in particular. Let's go through the code!

```
1.  #!/usr/bin/env perl
2.  use strict;
3.  use warnings;
4.
5.  use Data::Dumper::Compact qw(ddc);
6.  use List::Util qw(shuffle);
7.  use MIDI::Praxis::Variation qw(transposition);
8.  use MIDI::Util;
9.  use Music::Interval::Barycentric qw(cyclic_permutation);
```

```

10. use Music::Scales qw(get_scale_MIDI);
11. use Music::VoiceGen;

```

Here the standard perl preamble starts things off, followed by a handful of modules with methods and functions to use below. Next up is to define a few crucial parameters (that can be provided on the command-line):

```

1. my $bars = shift || 32;
2. my $bpm = shift || 180;
3. my $note = shift || 'B';
4. my $scale = shift || 'major';

```

Except for the number of bars, these are based on Reich's piece.

So this program uses global variables throughout. This is not a recommended practice for mission-critical, production software! Anyway, one of these globals is a voice generator created from the excellent [Music::VoiceGen](#) module:

```

1. my $octave = 2;
2. my @pitches = (
3.     get_scale_MIDI($note, $octave, $scale),
4.     get_scale_MIDI($note, $octave + 1, $scale),
5.     get_scale_MIDI($note, $octave + 2, $scale),
6. );
7. my $voice = Music::VoiceGen->new(
8.     pitches => \@pitches,
9.     intervals => [qw(-4 -3 -2 -1 1 2 3 4)],
10. );

```

This allows a random pitch to be generated based on three octaves and the given legal interval jumps. Next, a shuffled clarinet motif is defined and a clarinet note array declared:

```

1. my $cmotif = [shuffle qw(dhn en en en en)];
2. my @cnotes;

```

These will be used below, in the clarinet phrase generators... Next the notes for the piano tracks are generated with this code:

```

1. my $pmotif = ['en'] x 10;
2. my @pnotes;
3. my @transp;
4. for my $n (0 .. $#pmotif) {
5.     my $note = note_or_rest($n, $pmotif, \@pnotes);
6.     push @pnotes, $note;

```

```

7.         if ($note eq 'r') {
8.             push @transp, 'r';
9.         }
10.        else {
11.            my @transposed = transposition(-12, $note);
12.            push @transp, $transposed[0];
13.        }
14.    }

```

Here the piano motif is defined as ten eighth notes. Notes are added to the `@pnotes` and `@transp` arrays in a loop over the `$pmotif` and the function `note_or_rest()` is called. This function returns – you guessed it – either a note (as a MIDI pitch number) or a rest (as an 'r' character). This note is added to the `@pnotes` array, and a transposed version is added to the `@transp` array.

Ok. With all those things defined and populated, the next thing is to setup the MIDI stuff:

```

1.    my $volume = 98;
2.    my $pan = 10; # control change number
3.    my $pan_left = 32;
4.    my $pan_right = 86;
5.    my $score = MIDI::Util::setup_score(
6.        lead_in    => 0,
7.        signature  => '5/4',
8.        bpm        => $bpm,
9.        volume     => $volume,
10.    );

```

And synchronize the piano, violin and clarinet parts to play:

```

1.    $score->synch(
2.        \&piano1,
3.        \&piano2,
4.        \&violin1,
5.        \&violin2,
6.        \&clarinet1,
7.        \&clarinet2,
8.    );

```

Finally, as far as the execution of the program goes, a MIDI file, named after the program, is written to disk:

```

1.    $score->write_score("$0.mid");

```

Now for the subroutines! Let's consider the first one we encountered above,

`note_or_rest()`:

```
1. sub note_or_rest {
2.     my ($n, $motif, $notes) = @_;
3.     if (
4.         # We're at the end of the motif and the first note is a rest
5.         ($n == $#motif && $notes->[0] eq 'r')
6.         ||
7.         # The previous note is a rest
8.         (defined $notes->[$n - 1] && $notes->[$n - 1] eq 'r')
9.     ) {
10.         $note = $voice->rand;
11.     }
12.     else {
13.         $note = int(rand 10) <= 3 ? 'r' : $voice->rand;
14.     }
15.     return $note;
16. }
```

Here, a note is generated so that two rests are not in a row. That is what the if condition says basically. Otherwise either a rest or a note is generated based on a probability (i.e. return a rest approximately 40% of the time).

The companion to this function is the following, which actually adds either a rest or notes to the score:

```
1. sub play_note_or_rest {
2.     my ($motif, $notes) = @_;
3.     if ($notes->[0] eq 'r') {
4.         $score->r($motif);
5.     }
6.     else {
7.         $score->n($motif, @$notes);
8.     }
9. }
```

In the generation of piano notes, if the first is a rest, so is the second. That is why this code only considers the first element.

All that remains are the subroutines that play the pianos, violins and clarinets – which makes up the majority of the program actually. The first piano looks like this:

```
1. sub piano1 {
2.     MIDI::Util::set_chan_patch($score, 0, 0);
3.     $score->control_change(0, $pan, $pan_left);
```

```

4.     print 'Piano 1.1: ', ddc(\@pnotes);
5.     print 'Piano 1.2: ', ddc(\@transp);
6.     for my $i (1 .. $bars) {
7.         for my $n (0 .. $#pmotif) {
8.             play_note_or_rest($pmotif->[$n], [$pnotes[$n],
           $transp[$n]]);
9.         }
10.    }
11. }

```

Here, the MIDI channel and patch are both set to zero. Then the pan is set to the defined left value. The actual notes are shown (with the `ddc()` function of the also excellent [Data::Dumper::Compact](#) module). Next the computed notes (or rests) are added to the score in a loop over the defined number of “\$bars.” For each bar, a note or rest is played for each MIDI duration element of the motif. *Voilà!*

The other subroutines defining the other instruments are *each* different. This is so that the composition is not stale and redundant. But they are left to the reader to explore in [the program](#).

Here is a composition based on three runs of *very very many* that I finally sort-of liked: [Reichified II](#)

FILED UNDER: MUSIC, SOFTWARE
TAGGED WITH: MIDI, PERL

Epistemologist-at-large

^ Top